

# 18. Vetores

DIM0320

2015.1

# Sumário

1 Motivação

2 Vetores: definição e manipulação

3 Exercícios

- 1 Motivação
- 2 Vetores: definição e manipulação
- 3 Exercícios

# A temperatura média da semana

## Assunto

Escrever um algoritmo que

- lê as temperaturas de 7 dias sucessivos  $t_1, \dots, t_7$ ; e
- calcula
  - ▶ a temperatura média da semana  $\bar{t}$  e
  - ▶ o desvio padrão  $\sigma$ .

$$\sigma = \sqrt{\frac{1}{n} \sum_{i=1}^n (t_i - \bar{t})^2}$$

com  $n = 7$

# Solução

```
1 algoritmo "Temp"
2 var t, t1, t2, t3, t4, t5, t6, t7: real
3   m, soma : real
4   dp, soma_qdif : real
5   i : inteiro
6 inicio
7   i <- 1
8   soma <- 0
9   repita
10    escreval("Nova temperatura em C?")
11    leia(t)
12    soma <- soma + t
13    escolha i
14    caso 1 t1 <- t
15    caso 2 t2 <- t
16    caso 3 t3 <- t
17    caso 4 t4 <- t
18    caso 5 t5 <- t
19    caso 6 t6 <- t
20    caso 7 t7 <- t
21    outrocaso escreval("Nao acontece")
22    fimescolha
23    i <- i + 1
24  ate i = 8
25  // media
26  m <- soma / 7
```

```
1 // desvio padrao
2 i <- 1
3 soma_qdif <- 0
4 repita
5   escolha i
6   caso 1 t <- t1
7   caso 2 t <- t2
8   caso 3 t <- t3
9   caso 4 t <- t4
10  caso 5 t <- t5
11  caso 6 t <- t6
12  caso 7 t <- t7
13  outrocaso escreval("Nao acontece")
14  fimescolha
15  soma_qdif <- soma_qdif + (t - m)^2
16  i <- i + 1
17  ate i = 8
18
19  dp <- raizq(soma_qdif / 7)
20
21  escreval(m, dp)
22 fimalgoritmo
```

# Comentário

## Desvantagens

- Algoritmo fixo para 7 temperaturas
- O tratamento com escolha parece pode ser simplificado
- A informação contida em  $t_1, \dots, t_7$  é do mesmo tipo
- Tem que reescrever o algoritmo se tivesse 10 temperaturas

## Introdução de uma nova estrutura de dados

- **Vetores** permitem aliviar os problemas de generalização encontrados
- Vão simplificar e normalizar o código fonte

- 1 Motivação
- 2 Vetores: definição e manipulação
- 3 Exercícios

## Definição (Vetor)

- Estrutura de dados
  - ▶ composta (i.e. armazena mais que 1 valor)
  - ▶ homogênea (i.e. os valores são todos do mesmo tipo)
- Um **vetor** armazena uma sequência finita de valores do mesmo tipo de dados.
- A sequência é composta por **células**, cada uma armazenando um valor.
- Cada célula é acessível através um **índice** (valor inteiro)
- Um vetor tem um índice de **início** e um índice de **fim**.
- O **tamanho** de um vetor é o número de células disponíveis.

# Sintaxe

## Declaração

- Na zona de declarações de variáveis, pode-se declarar um variável do tipo vetor:

`<nome>: vetor [<tamanho>] de <tipo>`

## Observações

- `<nome>` : identificador
- `<tamanho>` : faixa de valores como `[1..5]`
- `<tipo>` : real, caractere, inteiro, lógico (qualquer tipo de base)

# Uso básico

## Como valor

- `k <- v[1] + 1`

## Como espaço de armazenamento

- `<nome_vetor>[<expr>] <- <expr>`
- `v[2] <- 2 * 3`

```
1 algoritmo "Ex1"
2 var v : vetor [1..2] de inteiro
3     i : inteiro
4 inicio
5     leia(v[1])
6     v[2] <- 3 * v[1]
7     v[2 * 1] <- v[3 \ 2] * 3
8     i <- 1
9     enquanto i < 3 faça
10         escreva(v[i])
11         i <- i + 1
12     fimenquanto
13 fimalgoritmo
```

# Solução com vetores

```
1 algoritmo "Temp"
2 var t: vetor [1..7] de real
3     m, soma : real
4     dp, soma_qdif : real
5     i : inteiro
6 inicio
7     i <- 1
8     soma <- 0
9     repita
10        escreval("Nova temperatura em C?")
11        leia(t[i])
12        soma <- soma + t[i]
13        i <- i + 1
14 ate i = 8
15 // media
16 m <- soma / 7
17
18 // desvio padrao
19 i <- 1
20 soma_qdif <- 0
21 repita
22     soma_qdif <- soma_qdif + (t[i] - m) * (t[i] - m)
23     i <- i + 1
24 ate i = 8
25
26 dp <- raizq(soma_qdif / 7)
27
28 escreval(m, dp)
29 fimalgoritmo
```

# Exemplo da temperatura

```
1 algoritmo "Temp"
2 var t: vetor [1..7] de real
3     m, soma : real
4     dp, soma_qdif : real
5     i : inteiro
6 inicio
7     soma <- 0
8     para i de 1 ate 7 passo 1 faca
9         escreval("Nova temperatura em C?")
10        leia(t[i])
11        soma <- soma + t[i]
12 fimpara
13 // media
14 m <- soma / 7
15
16 // desvio padrao
17 soma_qdif <- 0
18 para i de 1 ate 7 faca // passo default = 1
19     soma_qdif <- soma_qdif + (t[i] - m) * (t[i] - m)
20 fimpara
21
22 dp <- raizq(soma_qdif / 7)
23
24 escreval(m, dp)
25 fimalgoritmo
```

# Escolher entre enquanto..faca, repita..ate ou para..faca ?

- Sabe o número de repetições a efetuar ?
  - ▶ Caso **sim**, use o para..faca
  - ▶ Case **não**
    - ★ Tem que executar pelo menos uma vez o corpo do laço ?
    - ★ Caso **sim**, use o repita..ate
    - ★ Caso **não**, use o enquanto..faca

## Observação

- O poder expressivo dos 3 laços é o mesmo
- Distinguir entre repita..ate e enquanto..faca é também uma questão de estilo e de raciocínio.

# Perguntas ?



<http://dimap.ufrn.br/~richard/dim0320>

- 1 Motivação
- 2 Vetores: definição e manipulação
- 3 Exercícios

# Revisão

Qual laço é adaptado à escrita dos programas tratando os seguintes problemas:

- ① o total a pagar numa caixa de supermercado
- ② a pesquisa do dia mais chuvoso de um ano
- ③ o cálculo do perímetro de um polígono
- ④ o cálculo da duração de uma emissão de televisão a partir das horas de início e de fim.

# Soma de vetores

## Assunto

Escreva um algoritmo que:

- defina um vetor de elementos inteiros de tamanho 100,
- leia valores de entrada para este vetor
- escreva
  - 1 o valor a soma dos elementos que ocupam as posições pares do vetor
  - 2 o valor da soma dos elementos que ocupam as suas posições ímpares.

# Produto escalar

## Assunto

Escreva um algoritmo para calcular o produto escalar de dois vetores.

- A entrada do algoritmo consiste de um número inteiro  $n$ , com  $n \leq 25$ , e de  $2n$  números reais  $x_1, x_2, \dots, x_n$  e  $y_1, y_2, \dots, y_n$
- A saída do algoritmo é o produto escalar  $\bar{x} \cdot \bar{y}$ , com  $\bar{x} = (x_1, x_2, \dots, x_n)$  e  $\bar{y} = (y_1, y_2, \dots, y_n)$

$$\bar{x} \cdot \bar{y} = \sum_{i=1}^n x_i y_i$$

# Manipulação de vetor

## Assunto

Escreva um algoritmo que leia valores para um vetor de 100 inteiros e calcule e escreva :

- o maior e o menor elemento lido,
- o percentual de números pares,
- a média dos elementos do vetor e
- o número de elementos do vetor menores que a média.